



TRAINING ON COMPUTABLE GENERAL EQUILIBRIUM MODELLING

MODEL AUTETA

A CLOSED ECONOMY WITH GOVERNMENT

This pedagogical material was developed by Véronique Robichaud and accompanies Lesson 6. The model used is an adaptation of the model AUTETA presented in: Decaluwé, B., A. Martens and L. Savard (2001), « La politique économique du développement et les modèles d'équilibre général calculable. Une introduction », Montréal, Presses de l'Université de Montréal, 524 p. Since the SAM has been changed, the simulation results cannot be compared.

Table of content

Hypotheses	2
Sets.....	2
Equations	3
Variables	5
Parameters	7
The social accounting matrix for AUTETA	8
Correspondence between the SAM and the model	9
Schema.....	10
GAMS Code	11
Value of parameters	18
Simulations	20

Hypotheses

- Model of a closed economy with government
- Four industries / commodities:
 - Agriculture;
 - Manufacturing;
 - Services;
 - Public administrations
- Two factors of production:
 - Labour (mobile across sectors)
 - Capital (fixed by sector)
- Two categories of households:
 - Salaried households;
 - Capitalists.

Sets

Industries and commodities

$$i, j \in I = \{AGR, MAN, SER, PUB\}$$

(*AGR*: agriculture, *MAN*: manufacturing, *SER*: services, *PUB*: public administrations)

Tradable commodities and industries (excluding public administrations)

$$tr \in TR \subset I = \{AGR, MAN, SER\}$$

(*AGR*: agriculture, *MAN*: manufacturing, *SER*: services)

Sub-set of tradable industries and commodities excluding services:

$$bns \in BNS \subset TR = \{AGR, MAN\}$$

Households

$$h \in H = \{SAL, CAP\}$$

(*SAL*: salaried, *CAP*: capitalists)

Equations

Production 30

1. $VA_j = v_j \cdot XS_j$ 4
2. $CI_j = i\omega_j \cdot XS_j$ 4
3. $VA_{tr} = A_{tr} \cdot LD_{tr}^{\alpha_{tr}} \cdot KD_{tr}^{1-\alpha_{tr}}$ 3
4. $LD_{tr} = \frac{\alpha_{tr} \cdot PVA_{tr} \cdot VA_{tr}}{W}$ 3
5. $KD_{tr} = \frac{(1 - \alpha_{tr}) \cdot PVA_{tr} \cdot VA_{tr}}{R_{tr}}$ 3
6. $LD_{PUB} = VA_{PUB}$ 1
7. $DI_{tr,j} = aij_{tr,j} \cdot CI_j$ 12

Income and savings 18

8. $YH_{SAL} = W \cdot \sum_j LD_j + TG$ 1
9. $YH_{CAP} = \lambda \cdot \sum_{tr} R_{tr} \cdot KD_{tr} + DIV$ 1
10. $YDH_h = YH_h - DTH_h$ 2
11. $SH_h = \psi_h \cdot YDH_h$ 2
12. $CTH_h = YDH_h - SH_h$ 2
13. $YF = (1 - \lambda) \cdot \sum_{tr} R_{tr} \cdot KD_{tr}$ 1
14. $SF = YF - DIV - TDF$ 1
15. $YG = \sum_{tr} TI_{tr} + \sum_h DTH_h + DTF$ 1
16. $TI_{tr} = tx_{tr} \cdot P_{tr} \cdot XS_{tr}$ 3
17. $DTH_h = tyh_h \cdot YH_h$ 2
18. $DTF = tyf \cdot YF$ 1
19. $SG = YG - G - TG$ 1

Demand 12

$$20. C_{tr,h} = \frac{\gamma_{tr,h} CTH_h}{PD_{tr}} \quad 6$$

$$21. INV_{tr} = \frac{\mu_{tr} IT}{PD_{tr}} \quad 3$$

$$22. DIT_{tr} = \sum_j DI_{tr,j} \quad 3$$

Prices 12

$$23. PVA_{PUB'} = W \quad 1$$

$$24. PCI_j = \frac{\sum_{tr} PD_{tr} DI_{tr,j}}{CI_j} \quad 4$$

$$25. P_j = \frac{PVA_j \cdot VA_j + PCI_j \cdot CI_j}{XS_j} \quad 4$$

$$26. PD_{tr} = (1 + tx_{tr}) P_{tr} \quad 3$$

Equilibrium 8

$$27. XS_{bns} = \sum_h C_{h,bns} + DIT_{bns} + INV_{bns} \quad 2$$

$$28. XS_{PUB'} = \frac{G}{P_{PUB'}} \quad 1$$

$$29. LS = \sum_j LD_j \quad 1$$

$$30. KS_{tr} = KD_{tr} \quad 3$$

$$31. IT = \sum_h SH_h + SF + SG \quad 1$$

Verification of the Walras' law

$$32. LEON = XS_{SER'} - \sum_h C_{h,SER'} - DIT_{SER'} - INV_{SER'} \quad 1$$

Total: 81

Variables

Volume variables (quantities) 47

$C_{tr,h}$:	Consumption of commodity tr by type h households	6
CI_j :	Total intermediate consumption of industry j	4
$DI_{tr,j}$:	Intermediate consumption of commodity tr in industry j	12
DIT_{tr} :	Total intermediate demand for commodity tr	3
INV_{tr} :	Final demand of commodity tr for investment purposes	3
KD_{tr} :	Industry tr demand for capital	3
KS_{tr} :	Capital supply in industry tr	3
LD_j :	Industry j demand for labour	4
LS :	Total labour supply	1
VA_j :	Value added of industry j	4
XS_j :	Output of industry j	4

Prices 19

P_i :	Price of commodity i (excluding taxes)	4
PCI_j :	Intermediate consumption price index of industry j	4
PD_{tr} :	Price of commodity tr (including taxes)	3
PVA_j :	Price of industry j value added	4
R_{tr} :	Rental rate of capital in industry tr	3
W :	Wage rate	1

Nominal variables (values) 22

CTH_h :	Consumption budget of type h households	2
DIV :	Dividends	1
DTF :	Receipts from direct taxation on firms' income	1
DTH_h :	Receipts from direct taxation on household h income	2
G :	Current public expenditures	1
IT :	Total investment	1
SF :	Business savings	1

SG :	Government savings	1
SH_h :	Savings of type h households	2
TG :	Public transfers to salaried households	1
TI_{tr} :	Receipts from indirect tax on commodity tr	3
YDH_h :	Disposable income of type h households	2
YF :	Business income	1
YG :	Government income	1
YH_h :	Income of type h households	2
Other variables		1
$LEON$:	Excess supply on the market for services	1
		<hr/>
		Total: 89

Exogenous variables (closures)		Number
DIV :	Dividends	1
G :	Current public expenditures	1
KS_{tr} :	Capital supply in industry tr	3
LS :	Total labour supply	1
P_{agr} :	Price of commodity agr – numéraire	1
TG :	Public transfers to salaried households	1
		<hr/>
		Total: 8


Parameters

- A_{tr} : Scale parameter (Cobb-Douglas – production function)
- $aij_{tr,j}$: Coefficient (Leontief - intermediate consumption)
- α_{tr} : Elasticity (Cobb-Douglas – production function)
- $\gamma_{tr,h}$: Share of commodity tr in type h household consumption budget
- io_j : Coefficient (Leontief – total intermediate consumption)
- λ : Share of capital income received by capitalists
- μ_{tr} : Share of commodity tr in total investment expenditures
- ψ_h : Average propensity to save of type h household
- tx_{tr} : Indirect tax rate on commodity tr
- ty^f : Direct tax rate on firms' income
- ty^h_h : Direct tax rate on household h income
- v_j : Coefficient (Leontief – value added)


The social accounting matrix for AUTETA

	FACTORS		AGENTS				INDUSTRIES				COMMODITIES				ACC.	TOTAL
	1.	2.	3.	4.	5.	6.	7.	8.	9.	10.	11.	12.	13.	14.	15.	(1 to 15)
1. Labour							300	100	200	150						750
2. Capital							100	150	100							350
3. Salaried households	750					50										800
4. Capitalists		210			70											280
5. Firms		140														140
6. Government			40	28	7						10	25	15			125
7. Agriculture											500					500
8. Manufacturing												625				625
9. Services													600			600
10. Public administrations														200		200
11. Agriculture			162	21			50	150	90	10					27	510
12. Manufacturing			108	84			20	150	90	25					173	650
13. Services			270	105			30	75	120	15						615
14. Public administrations						200										200
15. Accumulation			220	42	63	-125										200
Total (1 to 15)	750	350	800	280	140	125	500	625	600	200	510	650	615	200	200	

Correspondence between the SAM and the model



Schema



GAMS Code

```
$TITLE      MODEL AUTETA
$$TITLE     AUTARKY WITH GOVERNMENT

* Model of a closed economy with government producing 3 goods using
* 2 factors owned by 2 types of households.

* October 2016

* This GAMS code was prepared by Veronique Robichaud. This basic CGE model is
* taken from Decaluwe, B., A. Martens and L. Savard (2001), " La politique
* economique du developpement et les modeles d'equilibre general calculable.
* Une introduction ", Montreal, Presses de l'Universit  de Montreal, 524 p.

* CALIBRATION

* Sets definition

SETS I Industries and commodities
/ AGR agriculture
  MAN manufacturing
  SER services
  PUB public administrations /

TR(I) Tradable commodities and industries
/ AGR agriculture
  MAN manufacturing
  SER services /

BNS(TR) Goods
/ AGR agriculture
  MAN manufacturing /

H Households
/ SAL labour endowed households
  CAP capital endowed households /

ALIAS (i,j)
;

* Parameters definition

PARAMETERS

A(tr)          Scale parameter (Cobb-Douglas - production function)
aij(tr,j)      Coefficient (Leontief - intermediate consumption)
alpha(tr)      Elasticity (Cobb-Douglas - production function)
gamma(tr,h)    Share of commodity tr in type h household consumption budget
io(j)          Coefficient (Leontief - total intermediate consumption)
lambda         Share of capital income received by capitalists
mu(tr)         Share of commodity tr in total investment expenditures
psi(h)         Average propensity to save of type h household
tx(tr)         Tax rate on commodity tr
tyf            Direct tax rate on firms' income
tyh(h)         Direct tax rate on household h income
v(j)           Coefficient (Leontief - value added)
```

```

* Definition of variables for the base year
* Volume variables (quantities)
CO(tr,h)      Consumption of commodity tr by type h households
CIO(j)        Total intermediate consumption of industry j
DIO(tr,j)     Intermediate consumption of commodity tr in industry j
DITO(tr)      Total intermediate demand for commodity tr
INVO(tr)      Final demand of commodity i for investment purposes
KDO(tr)       Industry tr demand for capital
KSO(tr)       Capital supply in industry tr
LDO(j)        Industry j demand for labour
LSO           Total labour supply
VAO(j)        Value added of industry j
XSO(j)        Output of industry j

* Prices
PO(i)         Price of commodity i (excluding tax)
PCIO(j)       Intermediate consumption price index of industry j
PDO(tr)       Price of commodity tr (including tax)
PVAO(j)       Price of industry j value added
RO(tr)        Rental rate of capital in industry j
WO            Wage rate

* Nominal variables (values)
CTHO(h)       Consumption budget of type h households
DIVO          Dividends
DTFO          Receipts from direct taxation on firms' income
DTHO(h)       Receipts from direct taxation on household h income
GO            Current public expenditures
ITO           Total investment
SFO           Business savings
SGO           Government savings
SHO(h)        Savings of type h households
TGO           Public transfers to salaried households
TIO(tr)       Receipts from indirect tax on commodity tr
YDHO(h)       Disposable income of type h households
YFO           Business income
YGO           Government income
YHO(h)        Income of type h households
;

* Initial data
PARAMETER
SAM(*,*,*,*) SAM for AUTETA;

$CALL GDXXRW.EXE AUTETA.xlsx par=SAM rng=MCS!A1:R18 Rdim=2 Cdim=2
$GDXIN AUTETA.gdx
$LOAD SAM
$GDXIN

```

```

* Assignment of variables
DIVO      = SAM('AG', 'CAP', 'AG', 'FIRM');
ITO       = SAM('OTH', 'TOT', 'OTH', 'ACC');
SFO       = SAM('OTH', 'ACC', 'AG', 'FIRM');
SHO(h)    = SAM('OTH', 'ACC', 'AG', h);
YFO       = SAM('OTH', 'TOT', 'AG', 'FIRM');
YHO(h)    = SAM('OTH', 'TOT', 'AG', h);
GO        = SAM('I', 'PUB', 'AG', 'GVT');
TGO       = SAM('AG', 'SAL', 'AG', 'GVT');
DTHO(h)   = SAM('AG', 'GVT', 'AG', h);
DTFO      = SAM('AG', 'GVT', 'AG', 'FIRM');
TIO(tr)   = SAM('AG', 'GVT', 'I', tr);
YGO       = SAM('OTH', 'TOT', 'AG', 'GVT');
SGO       = SAM('OTH', 'ACC', 'AG', 'GVT');
** For some variables in volume, we first assign a temporary value from
** the SAM. Later on, we will divide this value by the corresponding price.
XSO(j)    = SAM('OTH', 'TOT', 'J', j);
CO(tr,h)  = SAM('I', tr, 'AG', h);
DIO(tr,j) = SAM('I', tr, 'J', j);
INVO(tr)  = SAM('I', tr, 'OTH', 'ACC');
KDO(tr)   = SAM('F', 'KD', 'J', tr);
LDO(j)    = SAM('F', 'LD', 'J', j);

* Prices
WO        = 1;
RO(tr)    = 1;
PO(i)     = 1;

* Computation of variables in volume
LDO(j)    = LDO(j)/WO;
KDO(tr)   = KDO(tr)/RO(tr);
XSO(j)    = XSO(j)/PO(j);

* For other volumes, we must first compute the price including taxes.
* First, the tax rate:
tx(tr)    = TIO(tr)/[PO(tr)*XSO(tr)];
PDO(tr)   = [1+tx(tr)]*PO(tr);

CO(tr,h)  = CO(tr,h)/PDO(tr);
INVO(tr)  = INVO(tr)/PDO(tr);
DIO(tr,j) = DIO(tr,j)/PDO(tr);

* Calibration of other variables
LSO       = SUM[j, LDO(j)];
KSO(tr)   = KDO(tr);
VAO(tr)   = LDO(tr)+KDO(tr);
PVAO(tr)  = {WO*LDO(tr)+RO(tr)*KDO(tr)}/VAO(tr);
VAO('pub') = LDO('pub');
PVAO('pub') = WO;
DITO(tr)  = SUM[j, DIO(tr,j)];
CIO(j)    = SUM[tr, DIO(tr,j)];
PCIO(j)   = SUM[tr, PDO(tr)*DIO(tr,j)]/CIO(j);
YDHO(h)   = YHO(h)-DTHO(h);
CTHO(h)   = YDHO(h)-SHO(h);

* Calibration of parameters
* Production (Cobb-Douglas and Leontief)
alpha(tr) = WO*LDO(tr)/{PVAO(tr)*VAO(tr)};
A(tr)     = VAO(tr)/{LDO(tr)**alpha(tr)*KDO(tr)**(1-alpha(tr))};
v(j)      = VAO(j)/XSO(j);
io(j)     = CIO(j)/XSO(j);
aij(tr,j) = DIO(tr,j)/CIO(j);

```

```

* Distribution parameters
gamma(tr,h)      = PDO(tr)*CO(tr,h)/CTHO(h);
lambda          = {YHO('cap')-DIVO}/SUM[tr,RO(tr)*KDO(tr)];
mu(tr)          = PDO(tr)*INVO(tr)/ITO;
psi(h)          = SHO(h)/YDHO(h);

* Income tax rates
tyh(h)          = DTHO(h)/YHO(h);
tyf             = DTFO/YFO;

* Parameters to be displayed in the output file
DISPLAY A, alpha, io, v, aij, gamma, psi, mu, lambda, tyh, tyf, tx;
*$EXIT

* MODEL

* Definition of variables

VARIABLES

* Volume variables (quantities)
C(tr,h)         Consumption of commodity tr by type h households
CI(j)           Total intermediate consumption of industry j
DI(tr,j)        Intermediate consumption of commodity tr in industry j
DIT(tr)         Total intermediate demand for commodity tr
INV(tr)         Final demand of commodity i for investment purposes
KD(tr)          Industry tr demand for capital
KS(tr)          Capital supply in industry tr
LD(j)           Industry j demand for labour
LS              Total labour supply
VA(j)           Value added of industry j
XS(j)           Output of industry j

* Prices
P(i)            Price of commodity i (excluding tax)
PCI(j)          Intermediate consumption price index of industry j
PD(tr)          Price of commodity tr (including tax)
PVA(j)          Price of industry j value added
R(tr)           Rental rate of capital in industry j
W              Wage rate

* Nominal variables (values)
CTH(h)          Consumption budget of type h households
DIV             Dividends
DTF             Receipts from direct taxation on firms' income
DTH(h)          Receipts from direct taxation on household h's income
G              Current public expenditures
IT             Total investment
SF             Business savings
SG             Government savings
SH(h)          Savings of type h households
TG             Public transfers to salaried households
TI(tr)         Receipts from indirect tax on commodity tr
YDH(h)         Disposable income of type h households
YF             Business income
YG             Government income
YH(h)          Income of type h households

* Other variable
LEON           Excess supply on the market for services
;

```

```

* Definition of equations

* Production
EQUATIONS
XSEQ(j)      Value added demand in industry j (Leontief)
CIEQ(j)      Total intermediate consumption demand in industry j (Leontief)
VAEQ(tr)     Cobb-Douglas between labour and capital
LDEQ(tr)     Demand for labour by industry tr
KDEQ(tr)     Demand for capital by industry tr
LDPEQ       Equivalence between VA and LD for sector PUB
DIEQ(tr,j)  Intermediate consumption of commodity tr by sector j
;

XSEQ(j)..   VA(j) =e= v(j)*XS(j);

CIEQ(j)..   CI(j) =e= io(j)*XS(j);

VAEQ(tr)..  VA(tr) =e= A(tr)*LD(tr)**alpha(tr)*KD(tr)**(1-alpha(tr)) ;

LDEQ(tr)..  W*LD(tr) =e= alpha(tr)*PVA(tr)*VA(tr);

KDEQ(tr)..  R(tr)*KD(tr) =e= (1-alpha(tr))*PVA(tr)*VA(tr);

LDPEQ..     LD('pub') =e= VA('pub');

DIEQ(tr,j).. DI(tr,j) =e= aij(tr,j)*CI(j);

* Income and savings
EQUATIONS
YHSEQ       Household income (workers)
YHCEQ       Household income (capitalists)
YDHEQ(h)    Disposable income
SHEQ(h)     Household h savings
CTHEQ(h)    Consumption budget
YFEQ        Firms income
SFEQ        Firms savings
YGEQ        Government income
TIEQ(tr)    Receipts from indirect taxation
DTHEQ(h)    Receipts from income taxes (households)
DTFEQ       Receipts from income taxes (firms)
SGEQ        Government savings
;

YHSEQ..     YH('sal') =e= W*SUM[j,LD(j)]+TG;

YHCEQ..     YH('cap') =e= lambda*SUM[tr,R(tr)*KD(tr)]+DIV;

YDHEQ(h)..  YDH(h) =e= YH(h)-DTH(h);
SHEQ(h)..   SH(h) =e= psi(h)*YDH(h);

CTHEQ(h)..  CTH(h) =e= YDH(h)-SH(h);

YFEQ..     YF =e= (1-lambda)*SUM[tr,R(tr)*KD(tr)];

SFEQ..     SF =e= YF-DIV-DTF;

YGEQ..     YG =e= SUM[tr,TI(tr)]+SUM[h,DTH(h)]+DTF;

TIEQ(tr)..  TI(tr) =e= tx(tr)*P(tr)*XS(tr);

```



```

DTHEQ(h) ..      DTH(h) =e= tyh(h)*YH(h);

DTFEQ..         DTF =e= tyf*YF;

SGEQ..         SG =e= YG-G-TG;

* Demand
EQUATIONS
CEQ(tr,h)      Household h consumption of commodity tr
INVEQ(tr)      Investment in commodity tr
DITEQ(tr)      Intermediate demand for commodity tr
;

CEQ(tr,h) ..   PD(tr)*C(tr,h) =e= gamma(tr,h)*CTH(h);

INVEQ(tr) ..   PD(tr)*INV(tr) =e= mu(tr)*IT;

DITEQ(tr) ..   DIT(tr) =e= SUM[j,DI(tr,j)];

* Prices
EQUATIONS
PVAPEQ         Equivalence between PVA and W for the public sector
PCIEQ(j)       Intermediate consumption price index
CPEQ(j)        Production costs for sector j
PDEQ(tr)       Price of commodity tr including tax
;

PVAPEQ..       PVA('pub') =e= W;

PCIEQ(j) ..    PCI(j)*CI(j) =e= SUM[tr,PD(tr)*DI(tr,j)];

CPEQ(j) ..     P(j)*XS(j) =e= PVA(j)*VA(j)+PCI(j)*CI(j);

PDEQ(tr) ..    PD(tr) =e= P(tr)*[1+tx(tr)];

* Equilibrium
EQUATIONS
PEQ(bns)       Domestic absorption
PPUBEQ         Equilibrium on the market for public services
WEQ            Labour market equilibrium
REQ(tr)        Capital market equilibrium
ITEQ           Investment-savings equilibrium
;

PEQ(bns) ..    XS(bns) =e= SUM[h,C(bns,h)]+DIT(bns)+INV(bns);

PPUBEQ..       XS('pub')*P('pub') =e= G;

WEQ..          LS =e= SUM[j,LD(j)];

REQ(tr) ..     KS(tr) =e= KD(tr);

ITEQ..         IT =e= SUM[h,SH(h)]+SF+SG;

* Other
EQUATIONS
WALRAS         Verification of the Walras law
;

WALRAS..       LEON =e= XS('ser')-SUM(h,C('ser',h))-DIT('ser')-INV('ser');

```

```

* Initialisation of variables

C.l(tr,h)      = CO(tr,h);
CI.l(j)       = CIO(j);
CTH.l(h)      = CTHO(h);
DI.l(tr,j)    = DIO(tr,j);
DIT.l(tr)     = DITO(tr);
DIV.l         = DIVO;
DTH.l(h)      = DTHO(h);
DTF.l         = DTFO;
G.l           = GO;
INV.l(tr)     = INVO(tr);
IT.l          = ITO;
KD.l(tr)      = KDO(tr);
KS.l(tr)      = KSO(tr);
LD.l(j)       = LDO(j);
LS.l          = LSO;
P.l(i)        = PO(i);
PCI.l(j)      = PCIO(j);
PD.l(tr)      = PDO(tr);
PVA.l(j)      = PVAO(j);
R.l(tr)       = RO(tr);
SF.l          = SFO;
SG.l          = SGO;
SH.l(h)       = SHO(h);
TG.l          = TGO;
TI.l(tr)      = TIO(tr);
VA.l(j)       = VAO(j);
W.l           = WO;
XS.l(j)       = XSO(j);
YDH.l(h)      = YDHO(h);
YF.l          = YFO;
YG.l          = YGO;
YH.l(h)       = YHO(h);

* Closures

* P(AGR) is the numéraire
P.fx('agr')   = PO('agr');
* Capital is sector specific
KS.fx(tr)     = KSO(tr);
* Total labour suply is fixed
LS.fx         = LSO;
* Dividends are exogenous
DIV.fx        = DIVO;
* Public expenditures are exogenous
G.fx          = GO;
* Public transfers to householdes are exogenous
TG.fx         = TGO;

* Model execution

MODEL AUTETA Autarky with government /ALL/;
AUTETA.HOLDFIXED=1;
SOLVE AUTETA USING CNS;

```

Value of parameters

Definition	Symbol	Value
Scale parameter – Cobb-Douglas	A_{agr}	1.755
	A_{man}	1.960
	A_{ser}	1.890
Intermediate consumption coefficient – Leontief	$a_{ij_{agr,agr}}$	0.503
	$a_{ij_{man,agr}}$	0.197
	$a_{ij_{ser,agr}}$	0.300
	$a_{ij_{agr,man}}$	0.403
	$a_{ij_{man,man}}$	0.396
	$a_{ij_{ser,man}}$	0.201
	$a_{ij_{agr,ser}}$	0.302
	$a_{ij_{man,ser}}$	0.297
	$a_{ij_{ser,ser}}$	0.401
	$a_{ij_{agr,pub}}$	0.202
	$a_{ij_{man,pub}}$	0.496
	$a_{ij_{ser,pub}}$	0.302
Elasticity – Cobb-Douglas	α_{agr}	0.750
	α_{man}	0.400
	α_{ser}	0.667
Share of commodity in the consumption budget of salaried households	$\gamma_{agr,sal}$	0.300
	$\gamma_{man,sal}$	0.200
	$\gamma_{ser,sal}$	0.500
Share of commodity in the consumption budget of capitalists households	$\gamma_{agr,cap}$	0.100
	$\gamma_{man,cap}$	0.400
	$\gamma_{ser,cap}$	0.500
Total intermediate consumption coefficient – Leontief	io_{agr}	0.195
	io_{man}	0.583
	io_{ser}	0.486
	io_{pub}	0.242
Share of capital income to capitalists	λ	0.600
Share of commodity in total investment expenditures	μ_{agr}	0.135
	μ_{man}	0.865
	μ_{serl}	
Propensity to save	ψ_{sal}	0.289
	ψ_{cap}	0.167
Value-added coefficient – Leontief	v_{agr}	0.800
	v_{man}	0.400
	v_{ser}	0.500
	v_{pub}	0.750

Definition	Symbol	Value
Indirect tax rate	tx_{agr}	0.020
	tx_{man}	0.040
	tx_{ser}	0.025
Direct tax rate on firms' income	tyf	0.050
Direct tax rate on households' income	tyh_{sal}	0.050
	tyh_{cap}	0.100

Simulations

SIM1: 25% decrease of indirect tax rates

Definition	Symbol	Initial value	Simulation	Variation (%)
PRICE				
• wage rate	W	1	1.003	0.266
Rental rate of capital				
• agriculture	R_{agr}	1	1.001	0.080
• manufacturing	R_{man}	1	0.987	-1.319
• services	R_{ser}	1	1.012	1.242
Price of value added				
• agriculture	PVA_{agr}	1	1.002	0.219
• manufacturing	PVA_{man}	1	0.993	-0.688
• services	PVA_{ser}	1	1.006	0.590
• public administrations	PVA_{pub}	1	1.003	0.266
Intermediate consumption price index				
• agriculture	PCI_{agr}	1.025	1.016	-0.878
• manufacturing	PCI_{man}	1.029	1.017	-1.125
• services	PCI_{ser}	1.028	1.017	-1.054
• public administrations	PCI_{pub}	1.031	1.018	-1.302
Price of commodity(excluding taxes)				
• agriculture (numéraire)	P_{agr}	1	1.000	-
• manufacturing	P_{man}	1	0.990	-0.951
• services	P_{ser}	1	0.998	-0.232
• public administrations	P_{pub}	1	0.999	-0.126
Price of commodity(including taxes)				
• agriculture	PD_{agr}	1.02	1.015	-0.490
• manufacturing	PD_{man}	1.04	1.020	-1.903
• services	PD_{ser}	1.025	1.016	-0.840
PRODUCTION AND FACTORS				
Output				
• agriculture	XS_{agr}	500	499.306	-0.139
• manufacturing	XS_{man}	625	621.028	-0.635
• services	XS_{ser}	600	603.888	0.648
• public administrations	XS_{pub}	200	200.252	0.126
Value added				
• agriculture	VA_{agr}	400	399.444	-0.139
• manufacturing	VA_{man}	250	248.411	-0.635
• services	VA_{ser}	300	301.944	0.648
• public administrations	VA_{pub}	150	150.189	0.126

Labour				
• agriculture	LD_{agr}	300	299.445	-0.185
• manufacturing	LD_{man}	100	98.419	-1.581
• services	LD_{ser}	200	201.947	0.974
• public administrations	LD_{pub}	150	150.189	0.126
• total	LS	750	750	0
Capital				
• agriculture	KD_{agr}	100	100.000	-
• manufacturing	KD_{man}	150	150.000	-
• services	KD_{ser}	100	100.000	-
• total		350	350	-
Total intermediate consumption				
• agriculture	CI_{agr}	97.519	97.383	-0.139
• manufacturing	CI_{man}	364.460	362.144	-0.635
• services	CI_{ser}	291.847	293.738	0.648
• public administrations	CI_{pub}	48.477	48.538	0.126
Intermediate consumption				
• agriculture	$DI_{agr,agr}$	49.020	48.952	-0.139
	$DI_{man,agr}$	19.231	19.204	-0.139
	$DI_{ser,agr}$	29.268	29.228	-0.139
• manufacturing	$DI_{agr,man}$	147.059	146.124	-0.635
	$DI_{man,man}$	144.231	143.314	-0.635
	$DI_{ser,man}$	73.171	72.706	-0.635
• services	$DI_{agr,ser}$	88.235	88.807	0.648
	$DI_{man,ser}$	86.538	87.099	0.648
	$DI_{ser,ser}$	29.268	29.228	0.648
• public administrations	$DI_{agr,pub}$	9.804	9.816	0.126
	$DI_{man,pub}$	24.038	24.069	0.126
	$DI_{ser,pub}$	14.634	14.653	0.126
INCOME AND SAVINGS				
Income				
• workers	YH_{sal}	800	801.994	0.249
• capitalists	YH_{cap}	280	279.606	-0.141
• firms	YF	140	139.737	-0.188
• government	YG	125	112.287	-10.170
Disposable income				
• workers	YDH_{sal}	760	761.894	0.249
• capitalists	YDH_{cap}	252	251.645	-0.141
Public transfers to salaried households				
• transfers	TG	50	50.000	-

Receipts from direct taxation				
• workers	DTH_{sal}	40	40.100	0.249
• capitalists	DTH_{cap}	28	27.961	-0.141
• firms	DTF	7	6.987	-0.188
Receipts from indirect taxation				
• agriculture	TI_{agr}	10	7.490	-25.104
• manufacturing	TI_{man}	25	18.454	-26.185
• services	TI_{ser}	15	11.297	-24.689
Savings				
• workers	SH_{sal}	220	220.548	0.249
• capitalists	SH_{cap}	42	41.941	-0.141
• firms	SF	63	62.750	-0.396
• government	SG	-125	-137.713	10.170
DEMAND				
Salaried households – consumption				
• agriculture	$C_{agr,sal}$	158.824	160.004	0.743
• manufacturing	$C_{man,sal}$	103.846	106.124	2.194
• services	$C_{ser,sal}$	263.415	266.309	1.099
• total budget	CTH_{sal}	540	541.346	0.249
Capitalists – consumption				
• agriculture	$C_{agr,cap}$	20.588	20.661	0.351
• manufacturing	$C_{man,cap}$	80.769	82.220	1.796
• services	$C_{ser,cap}$	102.439	103.162	0.705
• total budget	CTH_{cap}	210	209.704	-0.141
Intermediate demand				
• agriculture	DIT_{agr}	294.118	293.699	-0.142
• manufacturing	DIT_{man}	274.038	273.686	-0.128
• services	DIT_{ser}	234.146	234.418	0.116
Demand for investment purposes				
• agriculture	INV_{agr}	26.471	24.942	-5.775
• manufacturing	INV_{man}	166.346	158.997	-4.418
• total	IT	200	187.527	-6.237
Current public expenditure				
• government	G	200	200.000	-

SIM2: 25% decrease of indirect tax rates (\$G exogenous, G endogenous)

Definition	Symbol	Initial value	Simulation	Variation (%)
PRICES				
• wage rate	W	1	0.999	-0.092
Rental rate of capital				
• agriculture	R_{agr}	1	1.009	0.901
• manufacturing	R_{man}	1	1.027	2.703
• services	R_{ser}	1	1.012	1.214
Price of value added				
• agriculture	PVA_{agr}	1	1.002	0.155
• manufacturing	PVA_{man}	1	1.016	1.575
• services	PVA_{ser}	1	1.003	0.341
• public administrations	PVA_{pub}	1	0.999	-0.092
Intermediate consumption price index				
• agriculture	PCI_{agr}	1.025	1.019	-0.621
• manufacturing	PCI_{man}	1.029	1.022	-0.637
• services	PCI_{ser}	1.028	1.021	-0.671
• public administrations	PCI_{pub}	1.031	1.024	-0.688
Price of commodity(excluding taxes)				
• agriculture (numéraire)	P_{agr}	1	1.000	-
• manufacturing	P_{man}	1	1.002	0.248
• services	P_{ser}	1	0.998	-0.165
• public administrations	P_{pub}	1	0.998	-0.241
Price of commodity(including taxes)				
• agriculture	PD_{agr}	1.02	1.015	-0.490
• manufacturing	PD_{man}	1.04	1.033	-0.716
• services	PD_{ser}	1.025	1.017	-0.774
PRODUCTION AND FACTORS				
Output				
• agriculture	XS_{agr}	500	503.726	0.745
• manufacturing	XS_{man}	625	631.937	1.110
• services	XS_{ser}	600	605.218	0.870
• public administrations	XS_{pub}	200	188.805	-5.598
Value added				
• agriculture	VA_{agr}	400	402.981	0.745
• manufacturing	VA_{man}	250	252.775	1.110
• services	VA_{ser}	300	302.609	0.870
• public administrations	VA_{pub}	150	141.603	-5.598

Labour				
• agriculture	LD_{agr}	300	302.984	0.995
• manufacturing	LD_{man}	100	102.798	2.798
• services	LD_{ser}	200	202.614	1.307
• public administrations	LD_{pub}	150	141.603	-5.598
• total	LS	750	750	0
Capital				
• agriculture	KD_{agr}	100	100.000	-
• manufacturing	KD_{man}	150	150.000	-
• services	KD_{ser}	100	100.000	-
• total		350	350	-
Total intermediate consumption				
• agriculture	CI_{agr}	97.519	98.245	0.745
• manufacturing	CI_{man}	364.460	368.505	1.110
• services	CI_{ser}	291.847	294.385	0.870
• public administrations	CI_{pub}	48.477	45.763	-5.598
Intermediate consumption				
• agriculture	$DI_{agr,agr}$	49.020	49.385	0.745
	$DI_{man,agr}$	19.231	19.374	0.745
	$DI_{ser,agr}$	29.268	29.486	0.745
• manufacturing	$DI_{agr,man}$	147.059	148.691	1.110
	$DI_{man,man}$	144.231	145.832	1.110
	$DI_{ser,man}$	73.171	73.983	1.110
• services	$DI_{agr,ser}$	88.235	89.003	0.870
	$DI_{man,ser}$	86.538	87.291	0.870
	$DI_{ser,ser}$	29.268	29.486	0.870
• public administrations	$DI_{agr,pub}$	9.804	9.255	-5.598
	$DI_{man,pub}$	24.038	22.693	-5.598
	$DI_{ser,pub}$	14.634	13.815	-5.598
INCOME AND SAVINGS				
Income				
• workers	YH_{sal}	800	799.307	-0.087
• capitalists	YH_{cap}	280	283.702	1.322
• firms	YF	140	142.468	1.763
• government	YG	125	113.349	-9.321
Disposable income				
• workers	YDH_{sal}	760	759.342	-0.087
• capitalists	YDH_{cap}	252	255.331	1.322
Public transfers to salaried households				
• transfers	TG	50	50.000	-

Receipts from direct taxation				
• workers	DTH_{sal}	40	39.965	-0.087
• capitalists	DTH_{cap}	28	28.370	1.322
• firms	DTF	7	7.123	1.763
Receipts from indirect taxation				
• agriculture	TI_{agr}	10	7.556	-24.441
• manufacturing	TI_{man}	25	19.005	-23.980
• services	TI_{ser}	15	11.329	-24.473
Savings				
• workers	SH_{sal}	220	219.809	-0.087
• capitalists	SH_{cap}	42	42.555	1.322
• firms	SF	63	65.344	3.721
• government	SG	-125	-125.000	-
DEMAND				
Salaried households – consumption				
• agriculture	$C_{agr,sal}$	158.824	159.468	0.406
• manufacturing	$C_{man,sal}$	103.846	104.505	0.634
• services	$C_{ser,sal}$	263.415	265.239	0.693
• total budget	CTH_{sal}	540	539.532	-0.087
Capitalists – consumption				
• agriculture	$C_{agr,cap}$	20.588	20.963	1.821
• manufacturing	$C_{man,cap}$	80.769	82.427	2.053
• services	$C_{ser,cap}$	102.439	104.603	2.112
• total budget	CTH_{cap}	210	212.776	1.322
Intermediate demand				
• agriculture	DIT_{agr}	294.118	296.334	0.753
• manufacturing	DIT_{man}	274.038	275.189	0.420
• services	DIT_{ser}	234.146	235.375	0.525
Demand for investment purposes				
• agriculture	INV_{agr}	26.471	26.961	1.854
• manufacturing	INV_{man}	166.346	169.815	2.086
• total	IT	200	202.709	1.354
Current public expenditure				
• government	G	200	188.349	-5.826

SIM3: 25% decrease of indirect tax rates (SG exogenous, TG endogenous)

Definition	Symbol	Initial value	Simulation	Variation (%)
PRICE				
• wage rate	W	1	1.003	0.261
Rental rate of capital				
• agriculture	R_{agr}	1	1.000	-0.026
• manufacturing	R_{man}	1	1.009	0.931
• services	R_{ser}	1	1.003	0.322
Price of value added				
• agriculture	PVA_{agr}	1	1.002	0.190
• manufacturing	PVA_{man}	1	1.007	0.662
• services	PVA_{ser}	1	1.003	0.282
• public administrations	PVA_{pub}	1	1.003	0.261
Intermediate consumption price index				
• agriculture	PCI_{agr}	1.025	1.018	-0.758
• manufacturing	PCI_{man}	1.029	1.020	-0.861
• services	PCI_{ser}	1.028	1.019	-0.871
• public administrations	PCI_{pub}	1.031	1.021	-0.974
Price of commodity(excluding taxes)				
• agriculture (numéraire)	P_{agr}	1	1.000	-
• manufacturing	P_{man}	1	0.997	-0.252
• services	P_{ser}	1	0.997	-0.295
• public administrations	P_{pub}	1	1.000	-0.048
Price of commodity(including taxes)				
• agriculture	PD_{agr}	1.02	1.015	-0.490
• manufacturing	PD_{man}	1.04	1.027	-1.211
• services	PD_{ser}	1.025	1.016	-0.903
PRODUCTION AND FACTORS				
Output				
• agriculture	XS_{agr}	500	498.926	-0.215
• manufacturing	XS_{man}	625	626.666	0.266
• services	XS_{ser}	600	600.241	0.040
• public administrations	XS_{pub}	200	200.095	0.048
Value added				
• agriculture	VA_{agr}	400	399.140	-0.215
• manufacturing	VA_{man}	250	250.666	0.266
• services	VA_{ser}	300	300.120	0.040
• public administrations	VA_{pub}	150	150.071	0.048

Labour				
• agriculture	LD_{agr}	300	299.141	-0.286
• manufacturing	LD_{man}	100	100.668	0.668
• services	LD_{ser}	200	200.120	0.060
• public administrations	LD_{pub}	150	150.071	0.048
• total	LS	750	750	0
Capital				
• agriculture	KD_{agr}	100	100.000	-
• manufacturing	KD_{man}	150	150.000	-
• services	KD_{ser}	100	100.000	-
• total		350	350	-
Total intermediate consumption				
• agriculture	CI_{agr}	97.519	97.309	-0.215
• manufacturing	CI_{man}	364.460	365.432	0.266
• services	CI_{ser}	291.847	291.964	0.040
• public administrations	CI_{pub}	48.477	48.500	0.048
Intermediate consumption				
• agriculture	$DI_{agr,agr}$	49.020	48.914	-0.215
	$DI_{man,agr}$	19.231	19.189	-0.215
	$DI_{ser,agr}$	29.268	29.205	-0.215
• manufacturing	$DI_{agr,man}$	147.059	147.451	0.266
	$DI_{man,man}$	144.231	144.615	0.266
	$DI_{ser,man}$	73.171	73.366	0.266
• services	$DI_{agr,ser}$	88.235	88.271	0.040
	$DI_{man,ser}$	86.538	86.573	0.040
	$DI_{ser,ser}$	29.268	29.205	0.040
• public administrations	$DI_{agr,pub}$	9.804	9.809	0.048
	$DI_{man,pub}$	24.038	24.050	0.048
	$DI_{ser,pub}$	14.634	14.641	0.048
INCOME AND SAVINGS				
Income				
• workers	YH_{sal}	800	789.004	-1.375
• capitalists	YH_{cap}	280	281.015	0.363
• firms	YF	140	140.677	0.483
• government	YG	125	112.043	-10.365
Disposable income				
• workers	YDH_{sal}	760	749.554	-1.375
• capitalists	YDH_{cap}	252	252.914	0.363
Public transfers to salaried households				
• transfers	TG	50	37.043	-25.913

Receipts from direct taxation				
• workers	DTH_{sal}	40	39.450	-1.375
• capitalists	DTH_{cap}	28	28.102	0.363
• firms	DTF	7	7.034	0.483
Receipts from indirect taxation				
• agriculture	TI_{agr}	10	7.484	-25.161
• manufacturing	TI_{man}	25	18.753	-24.989
• services	TI_{ser}	15	11.221	-25.191
Savings				
• workers	SH_{sal}	220	216.976	-1.375
• capitalists	SH_{cap}	42	42.152	0.363
• firms	SF	63	63.643	1.021
• government	SG	-125	-125.000	-
DEMAND				
Salaried households – consumption				
• agriculture	$C_{agr,sal}$	158.824	157.412	-0.889
• manufacturing	$C_{man,sal}$	103.846	103.674	-0.166
• services	$C_{ser,sal}$	263.415	262.161	-0.476
• total budget	CTH_{sal}	540	532.578	-1.375
Capitalists – consumption				
• agriculture	$C_{agr,cap}$	20.588	20.765	0.857
• manufacturing	$C_{man,cap}$	80.769	82.056	1.593
• services	$C_{ser,cap}$	102.439	103.747	1.277
• total budget	CTH_{cap}	210	210.761	0.363
Intermediate demand				
• agriculture	DIT_{agr}	294.118	294.444	0.111
• manufacturing	DIT_{man}	274.038	274.428	0.142
• services	DIT_{ser}	234.146	234.332	0.079
Demand for investment purposes				
• agriculture	INV_{agr}	26.471	26.305	-0.627
• manufacturing	INV_{man}	166.346	166.508	0.098
• total	IT	200	197.771	-1.114
Current public expenditure				
• government	G	200	200.000	-